

## PART\_A.PAS

```

PROGRAM MasterMind;
(*****
*
*   Solution to part A of Trabalho Pratico   *
*   Introducao a Computacao, 2005/2006     *
*       - P. Stallinga, 13 June 2006       *
*
*****)

Var code: array[1..100] of char;
    repetition: boolean;
    numcolumns, numcolors, numtry: integer;

PROCEDURE GenerateCode;
Var i, j, num: integer;
    c: char;
    allowed: boolean;
begin
    Randomize;
    for i := 1 to numcolumns do
        begin
            repeat
                num := Random(numcolors);
                c := Chr(num+65);
                allowed := TRUE;
                if NOT repetition then
                    for j := 1 to i-1 do
                        if code[j]=c then
                            allowed := FALSE;
                until allowed=TRUE;
                code[i] := c;
            end;
        end;
end;

FUNCTION Verify(try: string): boolean;
Var i, j, placecorrect, colorcorrect: integer;
    copycode: array[1..100] of char;
begin
    placecorrect := 0;
    colorcorrect := 0;
    for i := 1 to numcolumns do
        copycode[i] := code[i];
    for i := 1 to numcolumns do
        if try[i]=copycode[i] then
            begin
                placecorrect := placecorrect + 1;
                { make sure it cannot count anymore for correct color: }
                try[i] := ' ';
                copycode[i] := '.';
            end;
    Verify := (placecorrect=numcolumns);
    writeln('Correct place: ',placecorrect);
    for i := 1 to numcolumns do
        for j := 1 to numcolumns do
            if try[i]=copycode[j] then
                begin
                    colorcorrect := colorcorrect + 1;
                    { make sure it cannot count twice for correct color: }
                    try[i] := ' ';
                    copycode[j] := '.';
                end;
    writeln('Correct color: ',colorcorrect);
end;

Var c: char;
    i: integer;
    won: boolean;
    try: string;

```

PART\_A.PAS

```
begin
  writeln('Number of columns: ');
  readln(numcolumns);
  writeln('Number of colors: ');
  readln(numcolors);
  writeln('Repetitions allowed (y/n): ');
  readln(c);
  repetition := UpCase(c)='Y';
  writeln('Maxium Tries: ');
  readln(numtry);

  GenerateCode;
  i := 0;
  won := FALSE;
  repeat
    i := i+1;
    write('Try ',i,' (A..',Chr(64+numcolors),') : ');
    readln(try);
    won := Verify(try);
  until (i=numtry) OR won;
  if won then
    writeln('You won!')
  else
    begin
      write('Sorry, you lost. The correct code was: ');
      for i := 1 to numcolumns do
        write(code[i]);
      writeln;
    end;
end.
```